############################################################################
# Copyright (c) 2019, QuantStack and Mamba Contributors                    #
#                                                                          #
# Distributed under the terms of the BSD 3-Clause License.                 #
#                                                                          #
# The full license is in the file LICENSE, distributed with this software. #
############################################################################

# libmamba cmake module
# This module sets the following variables in your project::
#
#   libmamba_FOUND - true if libmamba found on the system
#   libmamba_INCLUDE_DIRS - the directory containing libmamba headers
#   libmamba_LIBRARY - the library for dynamic linking
#   libmamba_STATIC_LIBRARY - the library for static linking
#   libmamba_FULL_STATIC_LIBRARY - the library for static linking, incl. static deps

@PACKAGE_INIT@

set(CMAKE_MODULE_PATH "${CMAKE_CURRENT_LIST_DIR};${CMAKE_MODULE_PATH}")

@LIBMAMBA_CONFIG_CODE@

include(CMakeFindDependencyMacro)
find_dependency(fmt)
find_dependency(spdlog)
find_dependency(tl-expected)
find_dependency(nlohmann_json)
find_dependency(yaml-cpp)
find_dependency(reproc++)

if(NOT (TARGET libmamba-dyn OR TARGET libmamba-static))
    include("${CMAKE_CURRENT_LIST_DIR}/@PROJECT_NAME@Targets.cmake")

    if (TARGET mamba::libmamba-dyn)
        get_target_property(@PROJECT_NAME@_INCLUDE_DIR mamba::libmamba-dyn INTERFACE_INCLUDE_DIRECTORIES)
        get_target_property(@PROJECT_NAME@_LIBRARY mamba::libmamba-dyn LOCATION)
    endif()

    if (TARGET mamba::libmamba-static)
        get_target_property(@PROJECT_NAME@_INCLUDE_DIR mamba::libmamba-static INTERFACE_INCLUDE_DIRECTORIES)
        get_target_property(@PROJECT_NAME@_STATIC_LIBRARY mamba::libmamba-static LOCATION)
    endif()
endif()
